Nuclear Engineering Laboratory
M O N T E C U C C O L I N O
technical report
Nuclear Reactor Physics
Department of Energy and Nuclear Engineering
and of Environmental Control (DIENCA)
University of Bologna

Doc. ID: LIN-R03.2006
Subject: 2D merging algorithm
Date: August 2006

Author: M. Frignani, G. Grasso
Phone: +39-051-6441708
e-mail: giacomo.grasso@mail.ing.unibo.it

Spatial merging algorithms in 2D PIC codes

In a previous work [1] two different merging techniques were presented to reduce the number of simulation particles in Paticle-In-Cell Monte-Carlo-Collisional (PIC-MCC) codes. At least two particles are needed to replace a set of $N > 2$ particles in a single cell preserving the charge fractions on the nodes. Few preliminary results will be here presented and algorithm listings will be enclosed to show the main differences between the two techniques.

The Author
M. Frignani, G. Grasso

The Supervisor
prof. Marco Sumini

Revisions List:

Date	Author

Spatial merging algorithms in 2D PIC codes

M. Frignani, G. Grasso

August 2006

Abstract

In a previous work [1] two different merging techniques were presented to reduce the number of simulation particles in Paticle-In-Cell Monte-Carlo-Collisional (PIC-MCC) codes. At least two particles are needed to replace a set of $N > 2$ particles in a single cell preserving the charge fractions on the nodes. Few preliminary results will be here presented and algorithm listings will be enclosed to show the main differences between the two techniques.

Contents

1 Introduction **3**

2 Dimensionless quantities and cell reference **4**

3 Centroid and centrifugal moment conservation **4**

4 Addition of polar moment conservation **6**

5 Concluding remarks **9**

1 Introduction

In [1] two merging techniques were proposed to reduce the number of simulation particles per cell in a PIC-MCC code in order to reduce the computational time with no lack of precision.

For bilinear weighting in 2D meshes for electrostatic PIC codes, the solution of the electric field is independent from the charge distribution on scales lower then the grid spacing; consequently, the same solution is obtained if a set of N particles is reduced to a set of N' particles giving the same charge fraction on the nodes of each competing cell. In 2D grids with first-order linear weighting, $N > 2$ particles can be reduced to only $N' = 2$ particles placed in the same cell and giving the same charge fractions on its nodes.

A first merging technique is based on first order moments conservation: in this case, assuming equally weighted merging particles, a system of 3 equations in 4 unknowns is obtained. Solving it parametrically on one of the unknowns and choosing it randomly (in the allowed interval given by the cell bounds), the two merging particles can be placed in

the cell giving the same nodes charge fractions of the starting set. The arbitrariness of one coordinate prevent from obtaining the same set if the limit case $N = 2$ is considered.

The second procedure was developed with the aim to obtain a closed system able to preserve the particle positions in the limit case. The number of equations was increased including second order moments in a pseudo-inertia analogy. In particular, the polar moment limits the interval of variability for the particle position; the drawback is that the wise decision of equally weighted particles is no more achievable. The arbitrariness is indeed shifted to the particles weights ratio which bounds their distance from the charge centroid along a fixed direction.

In the present work, two algorithms will be presented and some main results will be compared.

2 Dimensionless quantities and cell reference

In what follows, a dimensionless reference with origin in the low-left corner of the cell will be considered. For consistency with [1], the coordinates system will be labeled with (ξ, η); the resultant cell is a 1×1 square.

In figure 1, the set of N starting particles is depicted; each particle is labeled with a number and, if not-equally weighted particles will be considered for generality, their weight w_i will replace the corresponding number. The centroid of charge will be oftenly represented and labeled with G, while the 2 particles of the new set will be marked by the prime. Different point syles will be used.

For grater convenince, a shifted reference centered in G will be used in formulae and listings and denoted by (ζ, γ) coordinates. The reader can refer to figure 2 or to [1] for a better understaing of the varibles meaning.

3 Centroid and centrifugal moment conservation

The conservation conditions of the charge fractions on the cell nodes are equivalent to conservation of the centroid of charge (and mass) and of the centrifugal pseudo-inertia moment. As derived in [1], these lead to the following coordinates for the two new particles

$$\zeta_{2'} = -\zeta_{1'} \tag{1a}$$

$$\gamma_{1'} = \frac{I_{\zeta\gamma}}{\zeta_{1'}} \tag{1b}$$

$$\gamma_{2'} = -\frac{I_{\zeta\gamma}}{\zeta_{1'}}\,, \tag{1c}$$

where $\zeta_{1'}$ can be randomly chosen in the interval

$$\zeta_{1'}^{\min} = \max\left\{\frac{|I_{\zeta\gamma}|}{2\eta_{\mathrm{G}}}, \frac{|I_{\zeta\gamma}|}{2(1-\eta_{\mathrm{G}})}\right\} < \zeta_{1'} < \min\{\xi_{\mathrm{G}}, (1-\xi_{\mathrm{G}})\} = \zeta_{1'}^{\max}\,.$$

An extract of a Fortran 90 source code that implements the describe procedure is presented in listing 1. Both the coordinates and the weights are here to be considered as di-

Listing 1: Extract of a F90 source code for particles merging with charge centroid e centrifugal pseudo-inertia moment conservation.

```
[...]
! Coordinates of the two new particles (1,2)
    REAL, DIMENSION(1:2) :: xMerged, yMerged
! Weights of the two new particles
    REAL, DIMENSION(1:2) :: wMerged
! Centrifugal moment of the system of N particles
!   IxyN = rhoC-(rhoB+rhoC)(rhoC+rhoD)
    REAL :: IxyN
! Interval extremes for xMerged(1)
    REAL :: x1max, x1min
[...]
! Charge centroid
    xGN=SUM(w*x); yGN=SUM(w*y)
! Centrifugal Inertia Moment
    IxyN=SUM(w*(x-xGN)*(y-yGN))

! New weights = 1/2
    wMerged=0.5
! Random choice for xMerged in [x1min,x1max]
    x1min=MAX((ABS(IxyN)/yGN),(ABS(IxyN)/(1-yGN)))
    x1max=MIN((xGN),(1-xGN))
    CALL RANDOM_NUMBER(xMerged(1))
! Coordinates calculation in the centroid reference
    xMerged(1)=xMerged(1)*(x1max-x1min)+x1min
    xMerged(2)=-xMerged(1)
    yMerged(1)=IxyN/(xMerged(1))
    yMerged(2)=-yMerged(1)
! Shift of the particles coordinates in the cell reference
    xMerged=xGN+xMerged; yMerged=yGN+yMerged;
[...]
```

mensionless and normalized to unity. The choice of arrays and the use of intrinsic functions and implicit cycle calculation makes the code neat and very easy to read.

In figures 3 and 4 the routine has been tested starting from the same configuration of $N = 5$ particles. The solutions differ in the randomly chosen ζ coordinate for the particle $1'$ (remember the assumption $\zeta_{1'} > 0$). The test shows that the charge fractions on the nodes are correctly conserved, while great discrepancies arises in the second order moments: a relative error between 46% and 63% for the first case and between 0.2% and 80% in the second is obtained.

In figure 5 a more complex and general case of $N = 10$ randomly weighted starting particles is presented. The procedure doesn't lose reliability; the error on second order not conserved moments lies between 18% and 79%.

Finally, in figure 6, a blank test is presented: the substitution of $N = 2$ particles in $N' = 2$ new particles is presented to show that, even starting from equal weights, it's possible only to obtain an equivalent, but not identical, configuration.

4 Addition of polar moment conservation

Writing the system (1) in the general case of not equally weighted particles and adding to them the conservation of the polar moment $I_{\varsigma\varsigma} = \sum_{i=1}^{N} \omega_i(\zeta_i^2 + \gamma_i^2)$, the coordinates of the new set of $N' = 2$ particles are

$$\zeta_{1'} = \varrho_{1'} \cos\vartheta \tag{2a}$$
$$\gamma_{1'} = \varrho_{1'} \sin\vartheta \tag{2b}$$
$$\zeta_{2'} = -\frac{\omega_{1'}}{\omega_{2'}}\zeta_{1'} \tag{2c}$$
$$\gamma_{2'} = -\frac{\omega_{1'}}{\omega_{2'}}\gamma_{1'}\,, \tag{2d}$$

where ϑ is given by

$$\vartheta = \frac{1}{2}\arcsin\left(\frac{2I_{\zeta\gamma}}{I_{\varsigma\varsigma}}\right) \in \left[-\frac{\pi}{4}, \frac{\pi}{4}\right],$$

while

$$\varrho_{1'} = \sqrt{\frac{\omega_{2'}}{\omega_{1'}} I_{\varsigma\varsigma}}\,.$$

The variability interval of ϑ can be extended to $[-\pi/2.\pi/2]$ simply evaluating if

$$\cos\vartheta = \sqrt{\frac{I_{\gamma\gamma}}{I_{\varsigma\varsigma}}} \gtrless \frac{\sqrt{2}}{2} \qquad \left(\text{with } 0 \le \frac{I_{\gamma\gamma}}{I_{\varsigma\varsigma}} \le 1\right);$$

if the $>$ is verified, then the angle ϑ desn't varies, while if the $<$ condition is satisfied, then the interval of ϑ changes into $[-\pi/2, -\pi/4[\in]\pi/4, \pi/2]$ since ϑ is substituted by $\pm(\pi/2 - \vartheta)$; is $\sin\vartheta > 0$ the plus sign has to be considered and it's simply necessary to invert $\cos\vartheta$ and $\sin\vartheta$ in (2), mantaining both $\zeta_{1'}$ and $\gamma_{1'}$ positive valued; in the opposite case, also a change of sign is needed leading to the inversion of $\cos\vartheta$ with $-\sin\vartheta$ and of $\sin\vartheta$ with $-\cos\vartheta$ in order to obtain $\zeta_{1'} > 0$. Recalling [1], once calculated

$$\cos(2\vartheta) = \sqrt{1 - \left(\frac{2I_{\zeta\gamma}}{I_{\varsigma\varsigma}}\right)^2},$$

the sine and cosine functions of ϑ are given by

$$\cos\vartheta = \sqrt{\frac{1}{2}(1 + \chi\cos(2\vartheta))} \tag{3a}$$
$$\sin\vartheta = \kappa\sqrt{\frac{1}{2}(1 - \chi\cos(2\vartheta))} \tag{3b}$$

where $|\kappa| = |\chi| = 1$ and signs defined as

$\vartheta \in$	$[-\pi/2, -\pi/4]$	$[-\pi/4, 0]$	$[0, \pi/4]$	$[\pi/4, \pi/2]$
	$I_{\zeta\gamma} < 0$	$I_{\zeta\gamma} < 0$	$I_{\zeta\gamma} > 0$	$I_{\zeta\gamma} > 0$
	$I_{\gamma\gamma}/I_{\varsigma\varsigma} < 1/2$	$I_{\gamma\gamma}/I_{\varsigma\varsigma} > 1/2$	$I_{\gamma\gamma}/I_{\varsigma\varsigma} > 1/2$	$I_{\gamma\gamma}/I_{\varsigma\varsigma} < 1/2$
χ	$-$	$+$	$+$	$-$
κ	$-$	$-$	$+$	$+$

The arbitrariness, previously met on $\zeta_{1'}$, here affects the weight $\omega_{1'}$, or alteratively, $\varrho_{1'}$, i.e. the distance of $1'$ from the barycenter G of the system. The cell bounds limit the variability of $\omega_{1'}$ in the following interval:

$$\omega_{1'}^{\min} = \frac{I_{\varsigma\varsigma}}{(\varrho_{1'}^{\lim})^2 + I_{\varsigma\varsigma}} < \omega_{1'} < \frac{(\varrho_{2'}^{\lim})^2}{(\varrho_{2'}^{\lim})^2 + I_{\varsigma\varsigma}} = \omega_{1'}^{\max},$$

where $\varrho_{1'}^{\lim}$ and $\varrho_{2'}^{\lim}$ are given in the following summarizing table:

	$\vartheta \in [-\pi/2, 0]\ (I_{\zeta\gamma} < 0)$	$\vartheta \in [0, \pi/2]\ (I_{\zeta\gamma} > 0)$
$\varrho_{1'}^{\lim}$	$\min\left\{\frac{1-\xi_G}{\cos\vartheta}, -\frac{\eta_G}{\sin\vartheta}\right\}$	$\min\left\{\frac{1-\xi_G}{\cos\vartheta}, \frac{1-\eta_G}{\sin\vartheta}\right\}$
$\varrho_{2'}^{\lim}$	$\min\left\{\frac{\xi_G}{\cos\vartheta}, -\frac{1-\eta_G}{\sin\vartheta}\right\}$	$\min\left\{\frac{\xi_G}{\cos\vartheta}, \frac{\eta_G}{\sin\vartheta}\right\}$

An extract of a fortran code intended to the new particles positions starting from a set of N particles in the cell is given in listing 2. It arises evident the increased complexity of the code with respect of listing 1. Moreover, in this case the merging into two equally weighed particles is not always possible: in the code an `IF...ELSE` construct is used to evaluate the opportunity $\omega_{1'}^{\min} < \omega_{1'} = \omega_{2'} = 1/2 < \omega_{1'}^{\max}$, since it is considered always preferable.

In order to reduce the time cosumiing procedure of `IF` tests, it is possible to compact the first two `IF...ELSE` construct in an analytical calculation. The first test (line 21–27) is related to the definition of variability interval of ϑ. The (3) can be contracted in

$$\cos\vartheta = \sqrt{\frac{1}{2}\left(1 - (-1)^{\mathrm{nint}(\frac{I_{\gamma\gamma}}{I_{\varsigma\varsigma}})}\cos(2\vartheta)\right)}$$
$$\sin\vartheta = \frac{I_{\zeta\gamma}}{|I_{\zeta\gamma}|}\sqrt{\frac{1}{2}\left(1 - (-1)^{\mathrm{nint}(\frac{I_{\gamma\gamma}}{I_{\varsigma\varsigma}})}\cos(2\vartheta)\right)}$$

where $\mathrm{nint}(x)$ is intended to return the nearest integer of x: 0 for $I_{\zeta\gamma}/I_{\varsigma\varsigma} < 0.5$ and 1 for $I_{\zeta\gamma}/I_{\varsigma\varsigma} \geq 0.5$, no other chance being possible. The refered lines of the code can be substituted by

```
1  costheta=SQRT(.5*(1-((-1)**NINT(IyyN/IzzN))*cos2theta))
2  sintheta=signIxyN*SQRT(.5*(1+((-1)**NINT(IyyN/IzzN))*cos2theta))
```

The second `IF...END IF` construct of interest is pertinent to the limiting values for $\varrho_{1'}$ and $\varrho_{2'}$ (see lines 29–35). In this case, the summarizing table of 7 can be contracted making

Listing 2: Extract of a F90 source code for particles merging with charge centroid, centrifugal and polar pseudo-inertia moment conservation.

```
[...]
! Coordinates of the two new particles (1,2)
    REAL, DIMENSION(1:2) :: xMerged, yMerged
! Weights of the two new particles and thei ratio
    REAL, DIMENSION(1:2) :: wMerged, wRatio
! Distances of the merged particles from the charge centroid
    REAL, DIMENSION(1:2) :: rMerged
! Centrifugal and polar moments of the system of N particles
!       IxyN = rhoC-(rhoB+rhoC)(rhoC+rhoD)
!       IzzN = SUM(w*(x**2+y**2)
    REAL :: IxyN, IzzN, signIxyN
! Rotation angle theta cos, sin functions
    REAL :: cos2theta, costheta, sintheta
! Limits on distance of the particles from G due to cell bounds
! and consequent limits for wMerged(1)
    REAL :: r1lim, r2lim, w1min, w1max
[...]
    signIxyN=IxyN/ABS(IxyN)
! Rotation angle: -pi/4<theta<pi/4
    cos2theta=SQRT(1-(2.*IxyN/IzzN)**2)
    IF ((IyyN/IzzN).GE.(.5)) THEN
        costheta=SQRT(.5*(1+cos2theta))  ! >0 always
        sintheta=signIxyN*SQRT(.5*(1-cos2theta)) ! same sign of IxyN
    ELSE    ! Conversion to theta in [-pi/2,-pi/4] or [pi/4,pi/2]
        costheta=SQRT(.5*(1-cos2theta))
        sintheta=signIxyN*SQRT(.5*(1+cos2theta))
    END IF
! Calculation of the limiting distances from G
    IF (IxyN.GE.0) THEN
        r1lim=MIN((1-xGN)/costheta,-yGN/sintheta)
        r2lim=MIN(xGN/costheta,-(1-yGN)/sintheta)
    ELSE
        r1lim=MIN((1-xGN)/costheta,(1-yGN)/sintheta)
        r2lim=MIN(xGN/costheta,yGN/sintheta)
    END IF
! Weights limits calculation, depending on r1lim, r2lim
    w1min=IzzN/(r1lim**2+IzzN); w1max=r2lim**2/(r2lim**2+IzzN)
! Particles weights assignment (w1=w2 if possible)
    IF((.5.GT.w1min).AND.(.5.LT.w1max)) THEN
        wMerged=.5
    ELSE  ! Random choice of wMerged(1)
        CALL RANDOM_NUMBER(wMerged(1))
        wMerged(1)=w1min+wMerged(1)*(w1max-w1min)
        wMerged(2)=1-wMerged(1)
    END IF
! Coordinates calculation in the centroid reference
  ! Radiuses
    wRatio=wMerged(2)/wMerged(1)
    rMerged(1)=SQRT(wRatio*IzzN); rMerged(2)=SQRT(IzzN/wRatio)
  ! Coordinates
    xMerged(1)=rMerged(1)*costheta; yMerged(1)=rMerged(1)*sintheta
    xMerged(2)=-rMerged(2)*costheta; yMerged(2)=-rMerged(2)*sintheta
! Shift of the particles coordinates in the cell reference
    xMerged=xGN+xMerged; yMerged=yGN+yMerged;
[...]
```

use of the sign of $I_{\zeta\gamma}$. It can be observed that only the second terms inside the $\min\ldots,\ldots$ functions undergo an exchange and sign inversion for $I_{\zeta\gamma} \lessgtr 0$; therefore, their expressions can be rewritten as

$$\varrho_{1'} = \min\left\{\frac{1-\xi_{\mathrm{G}}}{\cos\vartheta}, \frac{\frac{1}{2}\left(1+\frac{I_{\zeta\gamma}}{|I_{\zeta\gamma}|}\right)-\eta_{\mathrm{G}}}{\sin\vartheta}\right\}$$

$$\varrho_{2'} = \min\left\{\frac{\xi_{\mathrm{G}}}{\cos\vartheta}, -\frac{\frac{1}{2}\left(1-\frac{I_{\zeta\gamma}}{|I_{\zeta\gamma}|}\right)-\eta_{\mathrm{G}}}{\sin\vartheta}\right\}.$$

Then, the `IF` construct can be simply substituted by

```
1  r1lim=MIN((1-xGN)/costheta,(0.5*(1+signIxyN)-yGN)/sintheta)
2  r2lim=MIN(xGN/costheta,-(0.5*(1-signIxyN)-yGN)/sintheta)
```

The last `IF...END IF` (lines 39–45) is related to the weight assignment and cannot be simplified since its more complex dependences.

In figure 7 the random solution obtained by the code is presented. As reported, different weights are assigned to the two particles; if the picture is compared with figure 3, it can be pointed out that the direction of the two particles and G with respect to the ξ axis is different. While applying the first procedure this distance can vary due to the randomly assigned $\zeta_{1'}$, now it is not possible: the two particles will assume different weights but they be always aligned on the same direction (compare figure 7 with figure 8). Trying to force the code assigning the same weight to both the particles makes the second particle escaping the cell left bound, as represented in figure 9, where only the $2'$ is visible.

In the more general case of $N = 10$ starting particles with random positions and weights presented in figure 10, equal weights are assigned to the $N' = 2$ new particles with no lack of precision in conserved quantities.

Finally the blank test of $N = 2$ particles is presented. If the starting particles have equal weights, then the same configuration is obtained: as in the case of figure 11, the only difference is in the inversion of the two particles, due to the assumption $\zeta_{1'}$ not necessarily true for the starting particles. When the starting particles have random weights, then the same configuration cannot be precisely obtained. When possbile, the merging technique tries to homogenize the particles weights leading to a symmetric configuration (as presented in figure 12), while if it isn't possible, the weights are not preserved but randomly extracted.

This second more complex procedure allows to preserve both centrifugal and polar moments but not the axial ones, with relative errors up to 90% for the considered starting sets.

5 Concluding remarks

Two spatial merging techniques have been here compared showing their advantages and drawbacks on the basis of few examples.

The first procedure is rather simple and effective: the charge fractions on the cell nodes are exactly preserved and the arbitrariness on one of the merging particles coordinates can

be easily solved extracting a random number and scaling and shifting it in order to obtain a random position among those available inside the cell.

The second procedure requires three additional checks in order to determine the rotation angle ϑ, the limits over the allowed distances of the new particles from the charge centroid due to cell bounds and the possibility to assign equal weights to the new particles. Even if the first two tests can be avoided with not much more complex assignment functions, the procedure remains more complex requiring many trigonometric calculations. On the other hand, the benefit is more conservative result and a almost perfect answer to the two particles test (even if useless for the merging technique purpose).

Looking at some of the tests performed, reported in figure 13 for equally weighted starting particles and in figure 14 for randomly weighted ones, the second procedure seems to give better results, but there's no physical and numerical reason to prefer it to the first one. On the contrary, the numerically simplier procedure and the possibility to obtain equally weighted particles whatever the starting set, make the authors prefer the first technique to the second one.

References

[1] M. Frignani and G. Grasso. Spatial particles merging in 2D PIC codes. Technical Report LIN-R02.2006, LIN, 2006.



Figure 1: Particle positions in an unitary cell with dimensionless coordinates and weights.



Figure 2: Position of the two merged particles in the normalized reference $(\xi, \eta) - (\zeta, \gamma)$.



Figure 3: Five equally weighted starting particles in random positions are merged into two equally weigthed particles precisely conserving charge fractions on the nodes.



Figure 4: The same starting configuration of figure 3 leads to a different positioning of the two merging particles due to the randomly chosen $\xi_{1'}$.



Figure 5: 5 randomly weighted particles.



Figure 6: Blank test of $N = 2$ into $N' = 2$ particles.



Figure 7: Five equally weighted starting particles in random positions are merged into two randomly weigthed particles precisely conserving charge fractions on the nodes and polar moment of the system.



Figure 8: Same condition presented in figure 7 with differently chosen random weight $\omega_{1'}$.



Figure 9: Five equally weighted starting particles in random positions are merged into two equally weigthed particles: the particle labeled as $2'$ falls outside of the cell since conservation of charge fractions on the nodes and polar moment of the system cannot assure the positioning of eqaully weighted particles.



Figure 10: Five randomly weighted starting particles in random positions are merged into two equally weigthed particles with conservation of charge fractions on the nodes and polar moment of the system.



Figure 11: Equally weighted starting particles perfectly substituted by the new particles; note the labels inversion.



Figure 12: Randomly weighted starting particles merged into two equally weighted ones; the orientation is preserved, while the weights are homogenized if possible.



Figure 13: Comparison between the first (on the left) and the second (on the right) procedure for 3 sets of $N = 3$, 5, and 7 equally weighted particles.



Figure 14: Comparison between the first (on the left) and the second (onthe right) procedure for 3 sets of $N = 3$, 5, and 7 randomly weighted particles.